A. Veneziani - Report in formato .doc del listato dei regolatori rotativi 
con annesso oggetto audio

Il filmato contiene solo un fotogramma sulla time line principale che contiene il seguente codice.  Tale codice serve solo a inizializzare alcune variabili che vengono usate dal codice abbinato ai movieclip delle manopole, e l’attivazione e il link ad un certo pezzo  musicale, importato nella libreria di un oggetto Sound.  Quando il brano musicale viene importanto nella libreria bisogna quindi operare una concatenazione sul brano musicale importato. Da notare che e’ stata applicata una tecnica per ripetere il brano in loop, sfruttando un evento che viene attivato solo quando il brano stesso termina;  in quel momento il brano viene rilanciato in play, realizzando cosi’ un loop.
// --------------------------------------------------------------------------------------

// A. Veneziani - Filmato che realizza regolatori rotativi di volume e bilanciamento

//                   

// ---------------------------------------------------------------------------------------

// inizializzazione di alcune variabili

angolo_prec = 0;

rotazione_prec = 0;

// impone la non visibilita' della linea di trascinamento

linea._visible = false;

// rileva gia' all'inizio la lunghezza della linea presente sullo stage

// che sara' poi oggetto di operazioni di scala

lunghezza_linea = linea._width;

// abbina ad un oggetto Sound un suono opportunamente importato...

Snd = new Sound();

Snd.attachSound("chitarra");

// ...e regola il volume a 0 e il bilanciamento a centrato...

Snd.setVolume(0);

Snd.setPan(0);

// ...infine fa' partire la musica....

Snd.start();

// funzione che da un valore di gradi in formato Flash (-/+ 180°)

// ricava il valore con riferimento ad un angolo giro (360°)

function rotazione_assoluta(gr)

{

    if (gr < 0)  gr = gr + 360;

    return(gr);


}

//------------  Ripetizione del suono (loop)  ----------------------

// abbina all'evento onSoundComplete (ossia termine del brano scelto)

// la funzione continua

Snd.onSoundComplete = continua;

// funzione che rilancia il play del brano ogni volta che esso termina

function continua()

{

   Snd.start(); 


}

Il codice per il controllo della rotazione, abbinato ad ogni singola manopola e’ simile a quello riportato qui sotto (in attesa di una analisi più approfondita si riporta solo il codice stesso commentato).  Si puo’ notare che l’ “aggancio” alla manopola viene realizzato tramite un click sulla manopola stessa.  Dopodiche’ e’ reso possibile anche uscire dall’area della manopola per ruotarla.    La rotazione avviene con un cambiamento di coordinate rispetto al centro della manopola stessa, che funge da riferimento.  Da notare che la manopola,nella versione attuale possiede un sistema di limitazione del movimento agli estremi (non ancora del tutto funzionante)  e che la rotazione viene segnalata anche da una linea estensibile, che permette di visualizzare meglio l’entita’ della stessa.  Specificamente il codice riportato qui sotto si riferisce alla manopola del volume.   Alcune righe di codice mettono in relazione la rotazione con il valore della quantita’ controllata e queste variano di caso in caso, ovviamente.  Il rilascio del mouse comporta la variazione del valore della variabile booleana trascina, mentre trascinando il mouse mentre il tasto sinistro e’ premuto, vengono effettuate le operazioni di rotazione opportune.  Il filmato contiene anche una linea che viene posizionata e strecciata opportunamente, per renderla della lunghezza opportuna.  Tale linea segnala meglio visivamente l’entita’ della rotazione effettuata e viene automaticamente tirata dal centro della manpola alla posizione del cursore.  Per individuare la rotazione, ad ogni nuovo movimento del mouse si individua la variazione di angolazione del cursore rispetto al centro della manopola che sta’ trascinando e dello stesso angolo si ruota la manopola stessa.  
Essendo necessari spesso dei confronti fra una situazione precedente e una successiva, e’ necessario spesso introdurre delle variabili che mantengano memoria degli angoli precedentemente assunti dalla manopola o dal cursore del mouse.
// al momento del click sullo movieclip (tutta area dello stage)...

onClipEvent(mouseDown)

{

   // controlla se il click avviene sull'area che effettivamente rappresenta
   // il movieclip

   if (this.hitTest(_root._xmouse, _root._ymouse, true))    

   {

       trascina = true;


   // pone un vertice della linea "linea" nel centro del movieclip


   // (nel nostro caso il centro e' l'origine delle coordinate)

       _root.linea._x = this._x;

       _root.linea._y = this._y;

   }

   // calcola le coordinate del click rispetto al centro

   deltax_prec = this._x - _root._xmouse;

   deltay_prec = this._y - _root._ymouse;

   // ...questo valore serve poi a a calcolare l'angolo dal cui

   // valore inizia il trascinamento...

   angolo_prec = Math.atan2(deltay_prec, deltax_prec); 
   // viene registrata anche la rotazione effettuata fino a quel momento...

   rotazione_prec = this._rotation;

}

// al rilascio del mouse....

onClipEvent(mouseUp)

{

   // la linea che segnala una rotazione in corso deve sparire....


   _root.linea._visible = false;

   // viene annullata ogni altra operazione di rotazione in atto...

   trascina = false;

}

// quando il mouse viene mosso (con pulsante sinistro premuto)....

onClipEvent(mouseMove)

{

   // se e' attivo una rotazione...

   if (trascina)

   {


   // si calcola un nuovo angolo a cui la rotazione e' arrivata...  


   deltax = this._x - _root._xmouse;


   deltay = this._y - _root._ymouse;


   angolo = Math.atan2(deltay, deltax);



  // si calcola la rotazione attuale della manopola, basandosi sulla 1) rotazione


  // precedentemente raggiunta,  2) l'angolo attuale formato per differenza del


  // nuovo angolo calcolato e il vecchio angolo, precedentemente rilevato.


  // (questi angoli vengono calcolati in base alla posizione del cursore del mouse


  //  rispetto al centro della manopola).


  rotazione = ((angolo - angolo_prec) * 360 / 6.283185) + rotazione_prec;


  // se la rotazione ha valori ammissibili per i limiti imposti alla rotazione


  // della manopola, tali valori vengono trasmessi alla manopola ed essa viene


  // ruotata di conseguenza...

      if (((_root.rotazione_assoluta(rotazione) > 0) && (_root.rotazione_assoluta(rotazione) < 270)))

           this._rotation =  rotazione;


  // essendo questa una manopola che regola il volume esso viene regolato di conseguenza


  // facendo una proporzione in base al valore massimo raggiungibile (270°), ove si ha


  // 100% di volume (valore setVolume pari a 100)


  _root.Snd.setVolume((_root.rotazione_assoluta(this._rotation) / 270) * 100);

       // ----  operazioni sulla linea di segnalazione della rotazione -----


   // rende la linea di segnalazione visibile...


   
_root.linea._visible = true;


   // calcola la distanza tra il centro della manopola e la posizione attuale del mouse...


   distanza = Math.sqrt(Math.pow(deltax,2) + Math.pow(deltay,2));


   // ruota la linea opportunamente in relazione all'angolo di rotazione attuale,


   // calcolato in precedenza...


   _root.linea._rotation =  angolo * 360 / 6.283185 + 180;


   // calcola la nuova scala da applcare alla lunghezza della linea, per


   // farla diventare opportunamente piu' lunga (o piu' corta).

       scx = Math.abs(distanza / _root.lunghezza_linea);


   // applica la scala calcolata alla linea...


   _root.linea._xscale = scx * 100;

   }

}

Abbinato alla manopola per il bilanciamento troviamo una riga di codice diverso dal segmento qui sopra:

….

       // *** settaggio del valore di bilanciamento (deve variare tra +100 e -100)


   // Il centro della rotazione e' 135°, essendo la rotazione completa di 270°   ***

       _root.Snd.setPan((_root.rotazione_assoluta(this._rotation) - 135) / 135 * 100);

….
In questo caso si fa’ riferimento alla posizione centrale perche’ la manopola e’ posizionata centralmente nella sua posizione standard.   
